



Hamming Code Error correction implementation and translation to RISC-V Assembly

Othmane AZOUBI

June 1, 2024

`othmane.azoubi@um6p.ma`

Preface

This report focuses on the second assignment of Computer Organization and Architecture under the supervision of Professor Mohammed ATIBI. Following thorough research from various sources listed at the end of this report, I have elaborated on Hamming error-correcting codes. This includes defining the tools and algorithms used to support my implementation choices in RISC-V Assembly.

I. Introduction

Definition

Richard Hamming developed the concept of Hamming codes, which are a class of linear error-correcting codes. Even though advanced methods are now available, Hamming's work was revolutionary. He was the first to take on the difficult task of detecting and fixing errors in data storage and transmission. This established the basis of preserving data integrity across noisy channels. In this report, we have the opportunity to look into Hamming code. It's a 1950 technique originally created by Richard Hamming to fix data mistakes.

Throughout this report, I'll explain how it works, and walk through a basic implementation in RISC V Assembly while explaining the algorithms used.

Backstory

When Bell Labs began developing the Bell Model V computer in the late 1940s, Richard Hamming, the creator of the Hamming codes, was a member of the team. The machine's cycle times were measured in seconds and it was powered by electromechanical relays. Punched paper tape, measuring around seven-eighths of an inch in width and with up to six holes per row, was used to insert input. On weekdays, the machine would stop if relay problems were found, alerting operators with flashing lights to fix the problem. But when operators weren't around on the weekends or during off-peak hours, the machine would carry on automatically to the next duty.

Hamming, who worked regularly on the weekends, was getting more and more irritated with having to completely restart his programs every time a mistake was found. "Damn it, if the machine can detect an error, why can't it locate the position of the error and correct it?" he said in an interview, expressing his anger. His determination to find a solution to the error-correction issue was motivated by this dissatisfaction. In the years that followed, he created a number of algorithms to deal with this problem. He presented the technique that is now known as Hamming code in 1950; it is still used in modern technologies like ECC memory.

II. Applications, Advantages and Disadvantages

Applications

1. Satellites:

Hamming codes are essential in satellite communication systems. They ensure that data transmitted through space remains accurate, even when there are potential interferences or distortions. By detecting and correcting errors, they help maintain the integrity of communication signals.

2. **Computer Memory:**

In computer memory systems, like ECC RAM modules, Hamming codes play a crucial role. They help identify and fix memory errors, making data storage and retrieval more reliable in computer systems.

3. **Modems:**

Hamming codes are used in modems, especially when transmitting data over noisy channels like telephone lines. They ensure the accuracy of data transfer by detecting and correcting errors caused by signal distortions.

4. **PlasmaCAM:**

For systems like PlasmaCAM, which rely on precise cutting instructions, Hamming codes can be valuable. They help identify and correct errors in the instructions sent to the system, ensuring accurate cutting operations.

5. **Open Connectors:**

In systems using open connectors for data transmission, such as industrial automation or networking equipment, Hamming codes maintain data integrity. They detect and correct errors that may occur during data transmission, ensuring reliable communication.

6. **Shielding Wire:**

Hamming codes find application in systems using shielding wire, like those in aerospace or automotive industries. They detect and correct data corruption or noise introduced along the wire, ensuring signal integrity.

7. **Embedded Processors:**

Embedded processors, found in various electronic devices, benefit from Hamming codes for error detection and correction. By integrating Hamming code algorithms, data integrity is maintained, enhancing system reliability and performance.

Advantages

- Hamming code is effective on networks where data streams are vulnerable to single-bit errors.
- Besides detecting bit errors, Hamming code also identifies the specific bit containing the error, allowing for correction.
- Hamming codes are well-suited for computer memory and single-error correction applications due to their ease of use.

Limitations

- Hamming code is only capable of detecting and correcting single-bit errors and cannot handle multiple-bit errors.
- For Hamming code to work correctly, a substantial number of extra data bits must be added to the original message, which increases both the transmission time and the bandwidth usage.
- Hamming code lacks security because it can only identify errors that occur accidentally during data transmission, not intentional mistakes.

III. Former definitions

Before we discuss how the Hamming code algorithm works, first we need to define some concepts, helper algorithms and tools to aid us:

1. Parity bits

Parity: If counting the number of ones in your binary sequence results to be an even number, we say that the total parity is even. Otherwise, we say it's odd

1	0	0	1
0	1	0	1
1	1	0	0
0	0	1	1

(a) Example of an even block

0	1	1	0
1	0	1	0
1	0	0	0
0	1	0	1

(b) Example of an odd block

Figure 1: Examples of parity

Parity bit: Is a bit in a block of binary, that helps us detect and correct errors in transmitted data. It always ensures that the total number of bits (including the parity bit itself) is always even.

2. Power of two check

A power of two is a number that has only a single bit in it. This could be easily proven by the fact that $2^k = \underbrace{2 \times 2 \times \dots \times 2}_{k \text{ times}}$ is equivalent to 1 (2^0) being bit shifted k times to the left $\underbrace{10\dots\dots 00}_k 2$. This observation leads us to conclude that $2^k - 1 = \underbrace{011\dots\dots 11}_k 2$.

Thus, $\forall k \in \mathbb{N} (2^k) \& (2^k - 1) = 0$. Determining whether a number is a power of two, will help us knowing the positions of parity bits since in Hamming code they are positioned in bits where the position is a power of two.

3. Error Code, Detection and Correction

It's the code that is made out of $C = (C_k \dots C_2 C_1 C_0)_2$, with C_i ($0 \leq i \leq k$) being the parity of the region that contains the i -th parity bit (the sub-block that skips i bits checks i bits after the parity bit).

1	1	0	1	1	0	0	1
0	1	0	0	1	1	0	0
1	1	0	1	1	0	0	1
1	1	0	1	0	1	0	1

(a) C_1 sub-block(checks 1 skips 1)

1	1	0	1	1	0	0	1
0	1	0	0	1	1	0	0
1	1	0	1	1	0	0	1
1	1	0	1	0	1	0	1

(b) C_2 sub-block(checks 2 skips 2)

Figure 2: 32-bit blocks with C_1 and C_2 sub-blocks highlighted

A received message is called single-bit erroneous when $C \neq 0$. We can locate the position of such bit by computing $C - 1$, and thus correct a received message X by inverting its $(C - 1)$ -th bit using: $X \oplus (1 \ll (C - 1))$.

4. Brian Kernighan's Algorithm

First, let's try to turn off the rightmost set bit of a number n . To do this, we will consider two cases:

- **n is even:** In this case, n is divisible by 2^k with $k \in \mathbb{N}$. n is written in the format $n = \dots \dots \underbrace{10 \dots 00}_k 2$, and $n - 1 = \dots \dots \underbrace{01 \dots 11}_k 2$.

We conclude that $n \& (n - 1)$ is n but with the rightmost set bit turned off.

- **n is odd:** In this case, $n - 1$ is n but with the rightmost set bit turned off. This can also be written as $n \& (n - 1)$.

Conclusion: The expression $n \& (n - 1)$ can be used to turn off the rightmost set bit of a number n .

For example, consider the number $52 = 0011\ 0100_2$:

$$\begin{array}{rcl} n & = & 0011\ 0100 \\ n - 1 & = & 0011\ 0011 \\ \hline n \&(n - 1) & = & 0011\ 0000 \end{array}$$

Figure 3: Example taken from cp-algorithms.com

We can count the number of '1' in n using the above expression. The idea is to keep turning off the rightmost set bit (after counting it), till $n = 0$. Here is an implementation:

```
1 int countSetBits(int n) {
2     int count = 0;
3     while (n) {
4         n = n & (n - 1);
5         count++;
6     }
7     return count;
8 }
```

This algorithm will help us determine the parity of a binary block by checking the parity of the returned "count".

IV. Hamming code algorithm (32bits)

In this section, I will use all the tools and algorithms explained above to construct the Hamming code algorithm. I will first provide the algorithm in pseudo-code, and then write its RISC-V counterpart. But first, I will give the definitions of the *get_message()* and *parity* functions (Section III.4) in RISC-V:

```
1 parity:
2     # n = a0 and return value in a7
3     li a7, 0
4     loop_parity:
5         beqz a0, end_loop_parity
6         addi a7, a7, 1
7         addi t0, a0, -1
8         and a0, a0, t0
9         j loop_parity
10
11     end_loop_parity:
12     andi a7, a7, 1
13     jr ra
14
15
16 get_message:
17     #matricule = a0, r = a1 and return value in a7
```

```

18  xori t0, a0, 0xFFFFFFFF
19
20  sll t1, t0, a1 # (msg << r)
21  li t3, 32
22  sub t2, t3, a1 # 32 - r
23  srl t2, t0, t2 # (msg >> (32 - r))
24  or t0, t1, t2 # msg = (msg << r) | (msg >> (32 - r))
25
26  li t3, 0x00FFFFFF
27  and a7, t0, t3
28  jr ra

```

1. Hamming map:

It's the process of taking 24bits of data and placing them in a 32bits message while keeping away from bits that represents powers of 2 (Sections III.1 and III.2). Let's consider the following example:




Figure 4: Example hamming map

The algorithm behind it is straight forward:

Algorithm 1 Hamming Map

```
1: function HAMMING_MAP(msg)
2:   hamming  $\leftarrow$  0
3:   bit_position  $\leftarrow$  0
4:   for i  $\leftarrow$  0 to 31 do
5:     if i  $\neq$  0 and (i & (i - 1))  $\neq$  0 then            $\triangleright$  skipping parity bit positions
6:       if msg & (1  $\ll$  bit_position) then
7:         hamming  $\leftarrow$  hamming | (1  $\ll$  31)
8:       end if
9:       bit_position  $\leftarrow$  bit_position + 1
10:    end if
11:    hamming  $\leftarrow$  hamming  $\gg$  1
12:  end for
13:  return hamming
14: end function
```

RISC-V Implementation:

```
1 hamming_map:
2   #a0 = data and return value in a7
3   li t0, 0 #current position in the data
4   li a7, 0
5
6   #a7 = hamming
7   #t0 = bit_position
8   #a0 = msg
9
10  li t1, 0
11  loop_map:
12
13    #Parity bits checking:
14    beqz t1, map_skip_parity # check if t1 == 0
15    addi t2, t1, -1
16    and t2, t1, t2
17    beqz t2, map_skip_parity # check if (t1 & (t1 - 1)) == 0
18
19    li t2, 1
20    sll t2, t2, t0
21    and t2, a0, t2 # t2 = msg & (1 << bit_position)
22    beq t2, x0, map_zero
23    li t2, 1
24    slli t2, t2, 31
25    or a7, a7, t2 # hamming = hamming | (1 << (bit_position))
26  map_zero:
27    addi t0, t0, 1 # bit_position++
28
29    j map_skip_parity_end
```



```

30     map_skip_parity:
31         li t2, 0x7FFFFFFF
32         and a7, a7, t2 # Forcing parity bits to 0
33     map_skip_parity_end:
34
35     srl a7, a7, 1 # Shift hamming to the right so we can process the
36                 ↪ next bit
37
38     addi t1, t1, 1
39     li t2, 32
40     bne t1, t2, loop_map
41     jr ra

```

2. Hamming encode:

It's the process of setting the parity bits in our previously obtained Hamming map. Where each parity bit is responsible over the parity of it's sub-block (Consider Figure 2 as an example), Here is a quick algorithm to achieve the encoding:

Algorithm 2 Hamming Encode

```

1: function HAMMING_ENCODE(map)
2:   code  $\leftarrow$  map                                     ▷ Initialize code with map
3:   p  $\leftarrow$  PARITY(map & 0x55555554)
4:   code  $\leftarrow$  code | p
5:   p  $\leftarrow$  PARITY(map & 0x66666664)
6:   code  $\leftarrow$  code | (p << 1)
7:   p  $\leftarrow$  PARITY(map & 0x78787870)
8:   code  $\leftarrow$  code | (p << 3)
9:   p  $\leftarrow$  PARITY(map & 0x7F807F00)
10:  code  $\leftarrow$  code | (p << 7)
11:  p  $\leftarrow$  PARITY(map & 0x7FFF0000)
12:  code  $\leftarrow$  code | (p << 15)
13:  return code
14: end function

```

RISC-V Implementation:

```

1 hamming_encode:
2     #a0 = map and return value in a7
3     mv t6, a0 # return
4     mv t5, a0 # map
5     mv t0, x0 # Parity bit
6
7     addi sp, sp, -4

```

```

8      sw ra, 0(sp)
9
10     li t3, 0x55555554
11     and a0, t5, t3 # map & 0x55555554
12     jal parity # parity(map & 0x55555554)
13     or t6, t6, a7
14
15     li t3, 0x66666664
16     and a0, t5, t3 # map & 0x66666664
17     jal parity # parity(map & 0x66666664)
18     slli a7, a7, 1
19     or t6, t6, a7
20
21     li t3, 0x78787870
22     and a0, t5, t3 # map & 0x78787870
23     jal parity # parity(map & 0x78787870)
24     slli a7, a7, 3
25     or t6, t6, a7
26
27     li t3, 0x7F807F00
28     and a0, t5, t3 # map & 0x7F807F00
29
30     jal parity # parity(map & 0x7F807F00)
31     slli a7, a7, 7
32     or t6, t6, a7
33
34     li t3, 0x7FFF0000
35     and a0, t5, t3 # map & 0x7FFF0000
36     jal parity # parity(map & 0x7FFF0000)
37     slli a7, a7, 15
38     or t6, t6, a7
39
40     lw ra, 0(sp)
41     addi sp, sp, 4
42
43     mv a7, t6
44     jr ra

```

3. Hamming decode

After receiving the message we compute the parity of all sub-blocks to create an Error code C like discussed in Section III.3 the algorithm behind it is also simple:

Algorithm 3 Hamming Decode

```
1: function HAMMING_DECODE(code)
2:    $c0 \leftarrow \text{PARITY}((code \& 0x55555555))$ 
3:    $c1 \leftarrow \text{PARITY}((code \& 0x66666666))$ 
4:    $c2 \leftarrow \text{PARITY}((code \& 0x78787878))$ 
5:    $c3 \leftarrow \text{PARITY}((code \& 0x7F807F80))$ 
6:    $c4 \leftarrow \text{PARITY}((code \& 0x7FFF8000))$ 
7:    $C \leftarrow c0 \text{ OR } (c1 \ll 1) \text{ OR } (c2 \ll 2) \text{ OR } (c3 \ll 3) \text{ OR } (c4 \ll 4)$ 
8:   if  $C \neq 0$  then
9:      $code \leftarrow code \oplus (1 \ll (C - 1))$ 
10:  end if
11:  return code
12: end function
```

RISC-V Implementation:

```
1 hamming_decode:
2   #a0 = code and return value in a7
3   mv t6, a0
4   li t5, 0 # Error code
5
6   li t0, 0x55555555
7   li t1, 0x66666666
8   li t2, 0x78787878
9   li t3, 0x7F807F80
10  li t4, 0x7FFF8000
11
12  and t0, t6, t0 # code & 0x55555555
13  and t1, t6, t1 # code & 0x66666666
14  and t2, t6, t2 # code & 0x78787878
15  and t3, t6, t3 # code & 0x7F807F80
16  and t4, t6, t4 # code & 0x7FFF8000
17
18  addi sp, sp, -4
19  sw ra, 0(sp)
20
21  mv a0, t0
22  jal parity
23  slli a7, a7, 0
24  or t5, t5, a7
25
26  mv a0, t1
27  jal parity
28  slli a7, a7, 1
29  or t5, t5, a7
30
31  mv a0, t2
```

```

32     jal parity
33     slli a7, a7, 2
34     or t5, t5, a7
35
36     mv a0, t3
37     jal parity
38     slli a7, a7, 3
39     or t5, t5, a7
40
41     mv a0, t4
42     jal parity
43     slli a7, a7, 4
44
45     or t5, t5, a7
46
47     lw ra, 0(sp)
48     addi sp, sp, 4
49
50     beqz t5, no_error # if C != 0, there is an error
51     li t1, 1
52     addi t0, t5, -1 # t0 = C - 1
53     sll t1, t1, t0 # 1 << (C - 1)
54     xor t6, t6, t1 # code = code ^ (1 << (C - 1))
55 no_error:
56
57
58     mv a7, t6
59     jr ra

```

4. Hamming unmap

It's the process of applying the error correction formula, as seen in Section III.3, and extracting the original 24-bit data from the 32-bit received message after correction. Here is the algorithm behind it:

RISC-V Implementation:

Algorithm 4 hamming_unmap

```
1: function HAMMING_UNMAP(map)
2:   hamming  $\leftarrow$  map ▷ Initialize hamming with map
3:   matricule  $\leftarrow$  0
4:   bit_position  $\leftarrow$  0
5:   for i  $\leftarrow$  0 to 31 do
6:     if  $((i+1)\&i) == 0$  then
7:       continue
8:     else
9:       if (hamming & (1  $\ll$  i)) then
10:        matricule  $\leftarrow$  matricule | (1  $\ll$  bit_position)
11:      end if
12:      bit_position  $\leftarrow$  bit_position + 1
13:    end if
14:  end for
15:  return matricule
16: end function
```

```
1 hamming_unmap:
2   # a0 = map and return value in a7
3   li t0, 0 # bit position
4   li a7, 0 # matricule
5   # a0 = hamming
6
7   li t1, 0 # index
8   loop_unmap:
9
10    # Parity bits checking:
11    addi t2, t1, 1
12    and t2, t1, t2
13    beqz t2, unmap_skip_parity # check if (t1 & (t1 - 1)) == 0
14
15    li t2, 1
16    sll t3, t2, t1 # t3 = hamming & (1 << i)
17    and t3, a0, t3
18    beqz t3, unmap_zero
19    li t3, 1
20    sll t3, t3, t0 # (1 << bit_position)
21    or a7, a7, t3 # matricule = matricule | (1 << bit_position)
22  unmap_zero:
23    addi t0, t0, 1 # bit_position++
24
25    unmap_skip_parity:
26
27    addi t1, t1, 1
28    li t2, 32
29    bne t1, t2, loop_unmap
30
31  jr ra
```

5. Output

```
Matricule: 1825641
Message to be sent: 0x00124B7F
Mapped message: 0x02493774
Sent message: 0x0249377C
Received message: 0x0249377D
Corrected message: 0x0249377C
Unmapped message: 0x00124B7F
Exited with error code 0
Stop program execution!
-----
```

Figure 5: Example hamming map

Sources

- <https://cp-algorithms.com/algebra/bit-manipulation.html>
- <https://www.youtube.com/watch?v=X8jsijhl1IA>
- <https://www.geeksforgeeks.org/hamming-code-in-computer-network/>
- <https://www.techtarget.com/whatis/definition/Hamming-code>
- https://en.wikipedia.org/wiki/Hamming_code
- <https://zoo.cs.yale.edu/classes/cs323/doc/Hamming.pdf>
- <https://www.naukri.com/code360/library/examples-for-hamming-code>